Finding the volume of the region in the first octant under $z=4-x^2,y=4-x^2$ I can visualize most regions but the intersection of cylinders always confuses me, and I don't know how to use Mathematica very well yet.
We know that the projection of the region $R$ is the $xy$-plane is the parabola $y=4-x^2$. Hence,
$$0 \leq y \leq 4 \\ 0 \leq x \leq \sqrt{4-y}$$
We also know that both cylinders intersect along the plane $z=y$. The subregions that this plane divides are symmetrical, right? Is $(1)$ accurate then?
$$\iiint_R dV= 2 \int_0^4 \int_0^{\sqrt{4-y}} \int_0^y dzdxdy \tag 1 $$
 A: Your reasoning is correct (and is definitely an interesting approach), although I personally find it less obvious than just computing
$$\int_0^2\int_0^{4-x^2}\int_0^{4-x^2}dydzdx$$
both when justifying and solving the integrals. You can check that they give the same result though.
Hope this helps.
A: 
I can visualize most regions but the intersection of cylinders always confuses me, and I don't know how to use Mathematica very well yet.


Numerical experiments
Given the following homogeneous solid:
$$R := \left\{ (x,\,y,\,z) \in \mathbb{R}^3 : z \le 4-x^2, \; y \le 4-x^2, \; x \ge 0, \; y \ge 0, \; z \ge 0 \right\},$$
to define it in Wolfram Mathematica 12.0 and then calculate its main features, you can write:
restrictions = {z < 4 - x^2, y < 4 - x^2, x > 0, y > 0, z > 0};
R = ImplicitRegion[restrictions, {x, y, z}];

Region[R]
RegionMeasure[R]
RegionCentroid[R]
MomentOfInertia[R, {0, 0, 0}]

obtaining:


256/15
{5/8, 12/7, 12/7}
{{131072/945, -16, -16}, {-16, 74752/945, -16384/315}, {-16, -16384/315, 74752/945}}

Then you could be interested in the intervals on which to supplement by hand:
Reduce[restrictions]


0 < x < 2 && 0 < z < 4 - x^2 && 0 < y < 4 - x^2

and then writing the respective integrals iterated:
V = Integrate[1, 
              {x, 0, 2}, {z, 0, 4 - x^2}, {y, 0, 4 - x^2}]

Integrate[{x, y, z}, 
          {x, 0, 2}, {z, 0, 4 - x^2}, {y, 0, 4 - x^2}] / V

Integrate[{{y^2 + z^2, -x y, -x z}, 
           {-y x, x^2 + z^2, -y z}, 
           {-z x, -z y, x^2 + y^2}}, 
          {x, 0, 2}, {z, 0, 4 - x^2}, {y, 0, 4 - x^2}]

we obtain confirmation of the above already calculated:

256/15
{5/8, 12/7, 12/7}
{{131072/945, -16, -16}, {-16, 74752/945, -16384/315}, {-16, -16384/315, 74752/945}}

